

N-CLOCK UHRKARTE

**N-CLOCK  
ECHTZEITUHR  
FÜR APPLE II RECHNER**

INHALTSÜBERSICHT

SEITE INHALT

1	EINLEITUNG.
2	EINBAU UND INBETRIEBNAHME DER N-CLOCK-KARTE. SCHALTER AUF DER UHRKARTE.
3	AUSWAHL DER BETRIEBSART. BETRIEBSMODUS N-CLOCK
4	BETRIEBSMODUS APPLE-CLOCK (MOUNTAIN HARDWARE).
5	STELLEN DER UHR.
7	KALIBRIEREN DER N-CLOCK. DIE 30 SEKUNDEN-JUSTAGE DER N-CLOCK. DIE STROMVERSORGUNG.
8	LESEN DER ZEITINFORMATION DURCH BASIC.
11	SUCHEN DER N-CLOCK-KARTE DURCH EIN BASIC-PROGRAMM.
12	DOS 3.3-FILES MIT DATUMAUTOMATIK.
14	PRODOS MIT ZEIT/DATUMAUTOMATIK. CP/M UND DIE N-CLOCK
15	LESEN DER N-CLOCK UNTER MASCHINENSPRACHE.
16	LESEN DER N-CLOCK UNTER UCSD-PASCAL.
22	LESEN DER N-CLOCK OHNE NUTZUNG DER BETRIEBSSOFTWARE.
24	INTERRUPT-BETRIEB DER N-CLOCK-UHRKARTE UNTER APPLESOFT-BASIC.
25	BEISPIELPROGRAMM FÜR DEN INTERRUPT-BETRIEB DER UHRKARTE.
27	DOS 3.3 IN VERBINDUNG MIT DEM INTERRUPT-BETRIEB. INTERRUPTS AUSBLENDEN.
28	ZEIT UND DATUM WÄHREND DES INTERRUPT-BETRIEBES ERMITTELN.

EINLEITUNG

Dieses Handbuch beschreibt Installation und Inbetriebnahme der Uhrkarte, die im Apple //e und Apple II+ Verwendung findet. Ihr Betrieb ist ohne Disketten- Betriebssystemsoftware möglich.

Die Karte erweitert den Apple II-Computer um die Fähigkeit, Echtzeit und Datum in Intervallen von einer Millisekunde bis zu 99 Jahren zu verarbeiten. Ihre Anwendungsmöglichkeiten sind praktisch unbegrenzt. In jeder Anwendung, die Zeitangaben oder das Datum benötigen, können Sie die N-Clock einsetzen.

Die N-Clock ermöglicht neben ihrer eigenen, leistungsfähigsten Betriebsart die Emulation verschiedener anderer Uhrkarten. Das Betriebsprogramm der N-Clock ermöglicht extrem einfache Programmierung und höchstmögliche Kompatibilität mit existierender Software, die andere Uhrkarten voraussetzt.



## N-CLOCK-UHRKARTE

## EINBAU UND INBETRIEBNAHME DER N-CLOCK-KARTE

Unterbrechen Sie die Stromzufuhr des Apple-Computers. Zur Sicherheit ziehen Sie dazu den Netzstecker.

Vermeiden Sie es grundsätzlich, die Goldkontakte der Uhrkarte zu berühren und prüfen Sie vor dem Einbau, ob die Goldkontaktzungen verschmutzt sind. Gegebenenfalls müssen sie vorsichtig mit Aceton gereinigt werden. Diese Vorsichtsmaßnahmen gelten auch bei anderen Interface- Karten.

Entfernen Sie den Deckel des Gerätes.

Sie sehen jetzt im Inneren des Gerätes 7 nebeneinanderliegende, von links nach rechts mit 1 bis 7 durchnummerierte Steckplätze. Beim Apple II+ sind es 8 nebeneinanderliegende, von links nach rechts mit 0 bis 7 durchnummeriert Steckplätze.

Setzen Sie die Uhrkarte in einen beliebigen Slot, außer Slot 0, des Apple-Computers ein. Die **nicht** mit Bauteilen bestückte Seite der Karte muß zum Netzteil des Apple II weisen. Slot 4 sollte bevorzugt werden, da einige Softwarepakete eine Uhrkarte in Slot 4 voraussetzen.

## SCHALTER AUF DER UHRKARTE

Neben dem Akku-Modul der Uhrkarte finden Sie eine Schalterreihe mit 6 Dip-Schaltern. Die Schalter ermöglichen die Kontrolle über Bildschirm-String-Format, Interrupt-Zulassung und mehr.

O N	1	SET Muß geschlossen sein, wenn die Uhr gestellt wird.
	2	ADJ Wird benutzt um die Zeit um + oder - 30 Sek. zu stellen.
	3	MOD Wählt die Darstellungsweise an.
	4	MOD Wählt die Darstellungsweise an.
	5	NMI Erlaubt Non-Maskable Interrupt.
	6	IRQ Erlaubt Interrupt Request.

## N-CLOCK-UHRKARTE

## AUSWAHL DER BETRIEBSART

Die N Clock hat 2 durch Schalter 3 und 4 auswählbare Betriebsmodi. Der leistungsfähigste dieser Betriebsarten ist der sowohl mit DOS als auch mit ProDOS kompatible N-CLOCK-Modus.

Die anderen Betriebsarten ermöglichen die Kompatibilität mit verschiedenen kommerziellen Softwarepaketen, die ältere Uhrkarten unterstützen und keine Modifikation zur direkten Benutzung der N-Clock ermöglichen.

Betriebsmodus	Schalter 3	Schalter 4	Ausgabeformat
N-CLOCK	OFF	OFF	Verschiedene Formate
APPLE CLOCK MOUNTAIN HARDW.	ON	OFF	MO/TT SS:MM:SS.OJJ

BETRIEBSMODUS N-CLOCK (SCHALTER 3 und 4 OFF)

Dies ist die leistungsfähigste Betriebsart. Sie kann die Zeit- und Datum-Information in verschiedenster Formatierung liefern. Das gewünschte Format wird durch "PRINT" eines einleitenden Zeichens ausgewählt. Die einleitenden Format-Zeichen sind befehlsleich mit der Thunderclock-Uhrkarte und somit auch vollkompatibel mit dem ProDOS-Betriebssystem.

In der folgenden Tabelle finden Sie das einleitende Format-Zeichen zwischen den Anführungszeichen. Jedes Format wird am Beispiel Freitag, dem 5.7.85, 15.30 Uhr, aufgezeigt.

" "	MO/TT ST:MI:SS.MJJ 07/05 15:30:27.385	Apple Clock-Modus
":"	M MO/TT/JJ ST:MI:SS 5 07/05/85 15:30:27	N-Clock-Modus
"%"	TAG MON TT ST:MI:SS PM FRE JUL 05 03:30:27	Thunder-Clock Applesoft
"&"	TAG MON TT ST:MI:SS AM FRE JUL 05 15:30:27	Thunder-Clock Applesoft
"#"	MO,TI,TT,SS,MI,SS 07,05,05,15,30,27	Thunder-Clock Applesoft
">"	TAG MON TT ST:MI:SS PM FRE JUL 05 03:30:27 PM	Thunder-Clock Integer-Basic
"<"	TAG MON TT ST:MI:SS PM FRE JUL 05 15:30:27 PM	Thunder-Clock Integer-Basic



## N-CLOCK-UHRKARTE

Die Anführungszeichen haben die Aufgabe, Doppelpunkte in einem Eingabestring zu ermöglichen. Die ersten 4 Formate sind für die Nutzung unter Applesoft vorgesehen und enthalten das einleitende Anführungszeichen. Die beiden letzten Formate sind für Integer-Basic vorgesehen und beginnen mit einem Leerzeichen statt eines Anführungszeichens.

Obwohl es zwei Formate gibt, welche die Zeit im AM/PM 12 Stunden-Format darstellen, arbeitet die N-CLOCK intern im 24 Stunden-Format.

BETRIEBSMODUS APPLE-CLOCK (SCHALTER 3 ON, SCHALTER 4 OFF)

Der Apple-Clock-Modus emuliert das Format der alten MOUNTAIN-HARDWARE-Apple Clock, um Software- Kompatibilität zu kommerziellen Programmen zu gewährleisten, die eine Mountain Hardware-Uhrkarte benötigen und andere Karten nicht zulassen. Diverse Signatur-Bytes im Betriebsprogramm der Karte werden von diesen Softwarepaketen erkannt und ermöglichen die Benutzung der N-CLOCK-Uhrkarte.

Die Mountain Hardware-Uhrkarte liefert die Zeit/Datum-Information in folgendem Format

MO/TT ST;MI;SS.MMM

Beachten Sie bitte die Semikolons, die Stunden, Minuten und Sekunden trennen. Die drei letzten Zeichen sind Millisekunden. Stunden werden im 24 Stunden-Format (00-23) dargestellt. Wochentage und Jahre werden nicht angezeigt. Die Mountain Hardware-Uhr stellt weder Tag- noch Wochen- noch Jahreswerte zur Verfügung.

Die N-Clock ersetzt die Millisekunden durch den Wochentag und das Jahr. Zusätzlich werden die Semikolons durch Doppelpunkte ersetzt. Man erhält die Standard-Zeitanzeige:

MO/TT ST:MI:SS.TJJ  
07/05 15:30:27.585

Damit die Doppelpunkte von einer INPUT-Anweisung des Applesoft-Interpreters akzeptiert werden, wird das o.g. Format von einem Anführungszeichen eingeleitet.

## N-CLOCK-UHRKARTE

## STELLEN DER UHR

Um die N-Clock-Uhrkarte zu stellen, muß Schalter 1 der Dip-Switch Gruppe auf ON stehen; alle anderen Schalter müssen auf OFF stehen. Sobald Zeit und Datum korrekt gesetzt sind, stellen Sie sicherheitshalber Schalter 1 in die OFF-Position zurück.

Das folgende Programm legt die aktuelle Einstellung der N-Clock fest und ermöglicht die Änderung von Uhrzeit und Datum.

Das Programm durchsucht jeden Steckplatz zwischen 1 und 7 nach einigen Signatur-Bytes des N-CLOCK-Betriebsprogrammes und fordert dann zur Eingabe der neuen Werte auf.

```
010 ST=0
020 FOR I=1 TO 7
030 AD=12*4096 + I*256
040 IF PEEK(AD)=8 AND PEEK(AD+1)=120 AND PEEK(AD+254)=178 THEN ST=I:I=7
050 NEXT
060 IF ST=0 THEN PRINT" KEINE N-CLOCK GEFUNDEN":GOTO 130
070 REM
080 REM JE NACH STECKPLATZ ERFOLGT EIN CALL AUF DIE SPEICHERSTELLE
090 REM ADRESSE, AN DER DIE STELLROUTINE GESPEICHERT IST
100 REM
110 ADRESSE= 49656+(ST-1)*256
120 CALL ADRESSE
130 END
```

Wenn Sie wissen, in welchem Slot die N-CLOCK steckt, können Sie die CALL-Anweisung auch direkt über die Tastatur eingeben.

SLOT 1	=	49656 ODER -15880 ODER \$C1F8
SLOT 2	=	49912 ODER -15624 ODER \$C2F8
SLOT 3	=	50168 ODER -15368 ODER \$C3F8
SLOT 4	=	50424 ODER -15112 ODER \$C4F8
SLOT 5	=	50680 ODER -14856 ODER \$C5F8
SLOT 6	=	50936 ODER -14600 ODER \$C6F8
SLOT 7	=	51192 ODER -14344 ODER \$C7F8



## N-CLOCK-UHRKARTE

Es erscheint eine Bildschirmzeile mit den aktuellen Werten der N-CLOCK.

J=85 M=07 T=08 W=01 H=14 M=47 S=44

J=\_ M= T= W= H= M= S=00

J = Jahr  
 M = Monat  
 T = Tag  
 W = Wochentag zwischen 0-6. 0=Sonntag.  
 H = Stunden  
 M = Minuten  
 S = Sekunden

In der nächsten Zeile müssen Sie die neuen Werte eintragen. Die momentane Position wird durch einen Unterstrich angezeigt. Sie müssen alle neuen Werte einsetzen. Beachten Sie bitte besonders den Wochentag (0=Sonntag, 1=Montag,...), da ProDOS aus diesen Angaben das Jahr ermittelt. Der neu einzugebende Wert für Sekunden ist auf 00 voreingestellt.

Sobald alle Felder gefüllt sind, erscheint rechts neben der Zeile ein Fragezeichen. Nachdem Sie RETURN drücken, wird die N-CLOCK gestellt und das Programm beendet.

Drücken Sie die ESC-Taste, beginnt die Eingabe von vorne.

Drücken Sie die Taste E, so wird das Stellen der Uhr beendet.

Nach der Einstellung der Uhr muß der Schalter 1 der Dip-Switch Gruppe auf der N-CLOCK wieder auf OFF gestellt werden.

## N-CLOCK-UHRKARTE

## KALIBRIEREN DER N-CLOCK

Die N-CLOCK-Uhrkarte besitzt eine Quartz-kontrollierte Zeitbasis, die mit 32.768 oszilliert. Diese Frequenz kann durch einen Trimmer um +/- 2 Hz verstellt werden. Die N-CLOCK ist auf 32.7680 Hz mit einer Abweichung von +/- 0.002 % Genauigkeit justiert. Der Quartz-Kristall gewährleistet eine Genauigkeit mit einer Abweichung von 0.005 % im Jahr bei konstanter Temperatur.

Wenn die N-CLOCK permanent vor- oder nachgeht, kann die Frequenz mit dem Trimmer verstellt werden. Dreht man mit einem kleinen Schraubendreher die Schraube des Trimmers vorsichtig im Uhrzeigersinn, arbeitet die Uhr schneller. Dreht man gegen den Uhrzeigersinn, wird sie langsamer.

## DIE 30 SEKUNDEN-JUSTAGE DER N-CLOCK

Die N-CLOCK-Uhrkarte kann mit Hilfe einer einfachen Schalterumstellung um +/- 30 Sekunden justiert werden. Durch momentanes Einschalten (ON !) von Schalter 2 (ADJ) der Dip-Schalterreihe werden die Sekunden auf 00 gesetzt; wenn die aktuelle Zeit >30 Sekunden war, wird zusätzlich eine Minute addiert. Bitte stellen Sie den Schalter sofort nach der 30 Sekunden-Einstellung auf die OFF-Position zurück.

## DIE STROMVERSORGUNG

Die N-CLOCK verfügt über einen angebauten Akku, der die Stromversorgung auch gewährleistet, wenn der Computer abgeschaltet wird oder der Strom ausfällt. Sie stellt automatisch fest, ob die Stromversorgung des Gerätes unterbrochen wird und schaltet dann auf den Akku um. Die Uhr arbeitet weiter, während der Akku die Stromversorgung aufrecht erhält.

Solange der Strom des Apple-Computers eingeschaltet ist, wird der Akku auch während des Betriebes aufgeladen. Ein vollkommen entladener Akku benötigt ca. 10 Stunden Betrieb unter Stromversorgung, um wieder vollständig geladen zu sein. Eine Überladung ist nicht möglich; unter normalen Betriebsbedingungen muß der Akku nicht überwacht werden.

Ein voll geladener Akku kann die Uhr ca. 8 Monate funktionsfähig erhalten, ohne daß der Computer in Betrieb ist. Um den Akku geladen zu halten, sollte der Apple-Computer ca. 4 Stunden im Monat eingeschaltet sein.



## N-CLOCK-UHRKARTE

## LESEN DER ZEITINFORMATIONEN DURCH BASIC

Es ist sehr leicht, Datum und Zeitinformationen von der N-CLOCK zu erhalten. Die N-CLOCK enthält ein Betriebsprogramm, daß die Information gut lesbar aufbereitet.

Der einfachste Weg, die aktuelle Zeit zu ermitteln, erfordert nur folgendes kurzes Programm, bei dem davon ausgegangen wird, daß sich die N-CLOCK in Steckplatz 4 des Apple befindet.:

```

010 REM
020 REM EIN -UND AUSGABE AUF DIE N-CLOCK IN SLOT 4 UMSCHALTEN.
030 REM
040 PRINT CHR$(4);"IN#4"
050 PRINT CHR$(4);"PR#4"
060 REM
070 REM DATUM UND ZEIT DURCH INPUT LESEN.
080 REM
090 INPUT A$
100 REM
110 REM EIN -UND AUSGABE AUF TASTATUR UND BILDSCHIRM UMSCHALTEN.
120 REM
130 PRINT CHR$(4);"IN#0"
140 PRINT CHR$(4);"PR#0"
150 REM
160 REM DIE FERTIG AUFBEREITETE DATUM/ZEIT INFORMATION BEFINDET SICH
170 REM IM STRING A$. DAS FORMAT ERGIBT SICH AUS DEM ANZEIGEMODUS,
175 REM DER PER SCHALTER GEWÄHLT WURDE DER DANN FOLGENDE
180 REM AUSDRUCK ZEIGT DAS FORMAT AUF DEM BILDSCHIRM.
190 REM
200 PRINT A$
210 END

```

Wenn Sie per Schalterstellung auf der N-CLOCK-Uhrkarte das N-CLOCK-Format gewählt haben, können Sie die Ausgabeformate selbst bestimmen. Ändern Sie Zeile 90 wie folgt und betrachten Sie beim folgenden RUN das Ergebnis.

```

090 INPUT ":" ;A$
090 INPUT " " ;A$
090 INPUT "%" ;A$
090 INPUT "&" ;A$

```

Die N-CLOCK verfügt zudem über ein Ausgabeformat, das statt einer Zeichenkette numerische Werte abgibt. Es handelt sich bei dieser Ausgabemethode um eine Emulation des THUNDERCLOCK - Formates und wird intern vom Apple-ProDOS-Betriebssystem benutzt. Ändern Sie Zeile 90 und 200 und starten Sie das Programm erneut durch RUN.

```

090 INPUT "#" ;M,W,T,SS,MI,SE
200 PRINT M,W,T,SS,MI,SE

```

Um eine durchlaufende Zeitanzeige zu erhalten, ändern Sie das oben aufgeführte Programm wie folgt ab. Das Programm wird abgebrochen, sobald eine Taste der Apple-Tastatur betätigt wird.

## N-CLOCK-UHRKARTE

```

010 REM
020 REM EIN- UND AUSGABE AUF DIE N-CLOCK IN STECKPLATZ 4 UMSCHALTEN.
030 REM
040 PRINT CHR$(4);"IN#4"
050 PRINT CHR$(4);"PR#4"
060 REM
070 REM DATUM UND ZEIT DURCH INPUT LESEN.
080 REM
090 INPUT A$
100 REM
110 REM EIN- UND AUSGABE AUF DEN BILDSCHIRM UMSCHALTEN.
120 REM
130 PRINT CHR$(4);"IN#0"
140 PRINT CHR$(4);"PR#0"
150 REM
160 REM DIE BILDSCHIRM-AUSGABEPOSITION WIRD FESTGELEGT, DAMIT DIE AUS-
170 REM GABE PERMANENT AUF DER GLEICHEN STELLE ERSCHEINT.
180 REM
190 VTAB 12:HTAB 10
200 REM
210 REM DIE FERTIG AUFBEREITETE DATUM/ZEIT-INFORMATION BEFINDET SICH IM
212 REM STRING A$. DAS FORMAT ERGIBT SICH AUS DEM ANZEIGENMODUS, DER PER DIP-
220 REM SCHALTER GEWÄHLT WURDE. DER JETZT FOLGENDE AUSDRUCK ZEIGT DAS
230 REM FORMAT AUF DEM BILDSCHIRM.
240 REM
250 PRINT A$
260 REM ES WIRD GEPRÜFT, OB EINE TASTE GEDRÜCKT WURDE. WENN NEIN, IST
270 REM DER INHALT DER SPEICHERSTELLE MIT DER ADRESSE -16384 < 128 UND DIE PRO-
280 REM GRAMMAUSFÜHRUNG VERZWEIGT ZU ZEILE 40. IST DER INHALT >127, DANN
290 REM WIRD DER TASTATURPUFFER GELÖSCHT UND DAS PROGRAMM BEEENDET.
300 REM
310 IF PEEK (-16384) < 128 THEN GOTO 40
320 POKE -16368,0
330 END

```

Wenn Sie Zeit und Datum anders auf dem Bildschirm arrangieren wollen, können Sie die MID\$-Funktion des BASIC-Interpreters benutzen, um Teilmengen aus dem String "auszuschneiden". Die folgenden Beispiele zeigen, wie Sie auf Teile aus den Zeit/Datum-Zeichenketten unter den Formaten N-CLOCK und APPLE-CLOCK zugreifen können.

N-CLOCK Format: T MO/TT/JJ ST:MI:SE

TA\$	=	LEFT\$(A\$,1)	WOCHENTAG ZWISCHEN 0-6
MOS\$	=	MID\$(A\$,3,2)	MONAT ZWISCHEN 01-12
TA\$	=	MID\$(A\$,6,2)	TAGESDATUM ZWISCHEN 01-31
JA\$	=	MID\$(A\$,9,2)	JAHR ZWISCHEN 00-99
ST\$	=	MID\$(A\$,12,2)	STUNDE ZWISCHEN 00-23



## N-CLOCK-UHRKARTE

MIS\$ = MID\$(A\$,15,2) MINUTE ZWISCHEN 00-59  
 SE\$ = RIGHT\$(A\$,2) SEKUNDE ZWISCHEN 00-59

APPLECLOCK Format: MO/TA ST:MI:SE.TJA

TA\$ = MID\$(A\$,16,1) WOCHENTAG ZWISCHEN 0-6  
 JA\$ = RIGHT\$(A\$,2) JAHR ZWISCHEN 00-99  
 MO\$ = LEFT\$(A\$,2) MONAT ZWISCHEN 01-12  
 TA\$ = MID\$(A\$,4,2) TAGESDATUM ZWISCHEN 01-31  
 ST\$ = MID\$(A\$,7,2) STUNDE ZWISCHEN 00-23  
 MIS\$ = MID\$(A\$,10,2) MINUTE ZWISCHEN 00-59  
 SE\$ = MID\$(A\$,13,2) SEKUNDE ZWISCHEN 00-59

Sollten Sie bei der fortlaufenden Zeitanzeige auf dem Bildschirm nur an den Sekunden interessiert sein, ändern Sie (im N-CLOCK-Modus) folgende Zeile des o.g. Programmes:

```
250 SE$=RIGHT$(A$,2):PRINT SE$
```

Starten Sie dieses Programm durch RUN und es werden nur noch die Sekunden angezeigt.

Wenn Sie die Zeichenkette als Ziffer benötigen, benutzen Sie einfach die VAL-Funktion von BASIC. Beispielsweise wäre der Wochentag im N-CLOCK Format  $WT=VAL(LEFT$(A$,1))$ .

Sobald dieser Wert numerisch in WT vorliegt, können Sie z.B den Wochentag im Rahmen Ihres Programmes ausschreiben. Beispiel:

```
10 ON WT GOTO 20,30,40,50,60,70,80,90
20 WT$="Sonntag":GOTO 90
30 WT$="Montag":GOTO 90
40 WT$="Dienstag":GOTO 90
50 WT$="Mittwoch":GOTO 90
60 WT$="Donnerstag":GOTO 90
70 WT$="Freitag":GOTO 90
80 WT$="Samstag"
90 END
```

Sie können in ähnlicher Weise mit den Monaten verfahren oder der Jahreszahl eine 19 vorstellen. Die Möglichkeiten sind vielfältig.

## N-CLOCK-UHRKARTE

## SUCHEN DER N-CLOCK KARTE DURCH EIN BASIC PROGRAMM

Für jedes Ausgabeformat der N-CLOCK existiert eine eindeutige Signatur im Betriebsprogramm der Karte. Diese Signatur-Bytes befinden sich an genau definierten Speicherstellen, um Kompatibilität mit einigen kommerziellen Programmen zu gewährleisten, deren Software für andere Uhrkarten geschrieben wurde. Z.B. wurden spezielle Maßnahmen getroffen, damit der APPLECLOCK-Modus gegenüber der Software erscheint, als ob eine Mountain Hardware-Clock im Einsatz wäre.

Das folgende Programm sucht die Steckplätze nach einer N-CLOCK ab und meldet, in welchem Slot die Karte steckt oder ob keine Karte gefunden wurde. Zudem kann festgestellt werden, welches Ausgabeformat eingestellt ist.

```
10 ST=0
20 FOR I=1 TO 7
30 AD=12*4096 + I*256
40 IF PEEK(AD)=8 AND PEEK(AD+1)=120 AND PEEK(AD+254)=178 THEN ST=I:I=7
50 NEXT
60 IF ST=0 THEN PRINT" KEINE N-CLOCK GEFUNDEN":GOTO 90
70 A=PEEK(AD+255):PRINT "STECKPLATZ = ";ST
80 IF A=1 THEN PRINT"APPLECLOCK-FORMAT":GOTO90
85 IF A=3 THEN PRINT"N-CLOCK-FORMAT"
90 END
```



## N-CLOCK-UHRKARTE

## DOS 3.3-FILES MIT DATUMAUTOMATIK

Eine Anwendungsmöglichkeit besteht in dem automatischen Ergänzen des Datums bei jeder Datei, die im CATALOG angezeigt wird und durch SAVE, BSAVE oder RENAME abgespeichert wurde.

Einige Änderungen müssen am DOS 3.3-Betriebssystem vorgenommen werden, damit das Datum automatisch ergänzt wird. Zuerst ist eine Routine zu ergänzen, die bei jedem SAVE, BSAVE oder RENAME das Datum feststellt und den Dateinamen mit diesen Informationen überlagert.

Danach muß die DOS-Routine modifiziert werden, damit der Vergleich am rechten Ende des Namens, an dem die Datum-Information untergebracht ist, unterbleibt.

Wenn Sie das nachstehende Programm benutzen, um Ihr DOS 3.3 zu modifizieren, müssen Schalter 3 und 4 der Dip-Switch-Schaltergruppe auf OFF stehen. Das N-CLOCK-Ausgabeformat muß gewählt werden, da andere Uhrenformate nicht das Jahr liefern. Nach Programmende ist das DOS in Ihrem Speicher modifiziert. Auf Ihrer Diskette bleibt das DOS allerdings unverändert.

Sobald Ihr DOS 3.3 im Arbeitsspeicher modifiziert ist, können Sie durch das INIT-Kommando jede Diskette mit diesem neuen modifizierten DOS 3.3 versehen.

Der Steckplatz der N-CLOCK darf nicht verändert werden, da die Position der Karte im DOS vermerkt ist. Wenn der Platz der Karte gewechselt wird, müssen Sie die Installationsprozedur erneut durchspielen.

Die zugelassenen Längen der Dateinamen sind in 2 Adressen des DOS 3.3 gespeichert. Die Inhalte dieser beiden Adressen müssen an die tatsächliche Länge des Dateinamens angepasst werden. Der tatsächliche Dateiname ist 30 Zeichen lang, abzgl. der Länge der Datumsangabe MO/TA/JA. Somit ist der Dateiname auf 30-9 = 21 Zeichen beschränkt.

```

010 HOME:PRINT"DOS 3.3 MODIFIKATION ZUR AUTOMATISCHEN"
020 PRINT"DATUMS-SPEICHERUNG":PRINT:PRINT
030 REM
050 A = 193
060 B = 256 * A
100 IF PEEK (B) = 8 THEN GOTO 400
110 A = A + 1
120 IF A < 200 THEN GOTO 60
130 PRINT " N-CLOCK-KARTE NICHT GEFUNDEN "
140 END
REM*****
400 IF PEEK (B + 1) <> 120 THEN GOTO 110
410 IF PEEK (B + 252) <> 195 THEN GOTO 110
420 IF PEEK (B + 253) <> 204 THEN GOTO 110

```

## N-CLOCK-UHRKARTE

```

430 IF PEEK (B + 255) <> 3 THEN GOTO 110
440 PRINT" N-CLOCK UHRKARTE IN SLOT "A-192" GEFUNDEN ! "
450 PRINT: PRINT " DOS 3.3 WIRD NUN MODIFIZIERT"
500 POKE 42883, 20: POKE 44085, 209: POKE 44086, 182: POKE 45571,21: POKE 46792,2
510 POKE 46793, 6: POKE 46794, 7: POKE 46795, 8: POKE 46796,3: POKE 46797,4
520 POKE 46798, 5: POKE 46799, 9: POKE 46800, 10: POKE 46801,173: POKE 46802, 255
530 POKE 46803, A: POKE 46804,201: POKE 46805, 3: POKE 46806, 208: POKE 46807, 34
540 POKE 46808, 169: POKE 46809, 186: POKE 46810, 32: POKE 46811, 11: POKE 46812, A
550 POKE 46813, 32: POKE 46814, 8: POKE 46815, A: POKE 46816, 160: POKE 46817, 8
560 POKE 46818, 152: POKE 46819, 24: POKE 46820, 109 POKE 46821, 217: POKE 46822, 181
570 POKE 46823, 170: POKE 46824, 152: POKE 46825, 72 POKE 46826, 185: POKE 46827, 200
580 POKE 46828, 182: POKE 46829, 168: POKE 46830, 185 POKE 46831, 0: POKE 46832, 2
590 POKE 46833, 157: POKE 46834, 222: POKE 46835, 180 POKE 46836, 104: POKE 46837, 168
600 POKE 46838, 202: POKE 46839, 136: POKE 46840, 16 POKE 46841, 238: POKE 46842, 76
610 POKE 46843, 55: POKE 46844, 176
620 PRINT: PRINT
630 PRINT"DOS 3.3 KANN NUN PROGRAMME MIT DATUM ";
640 PRINT"WEGSPEICHERN.": PRINT
650 PRINT"BITTE VON DER MASTER-DISKETTE DOS 3.3"
660 PRINT"-HELLO- LADEN (LOAD HELLO) UND EINE"
665 PRINT" DISKETTE INITIALISIEREN (INIT HELLO)"
670 PRINT"DAS DOS DIESER DISKETTE SPEICHERT"
680 PRINT"PROGRAMME AUTOMATISCH MIT DATUM. "

```



## N-CLOCK-UHRKARTE

## PRODOS MIT ZEIT/DATUMAUTOMATIK

ProDOS ergänzt automatisch die Zeit und das Datum im N-CLOCK-Betriebsmodus (Schalter 3 und 4 auf OFF). ProDOS geht etwas weiter als die Modifikationen im DOS-Betriebssystem. Neben Datum und Zeit der Dateierstellung wird auch Datum und Zeit der letzten Modifikation festgehalten.

Beim Umgang mit ProDOS ist besondere Vorsicht notwendig. Die Zeitermittlung in ProDOS wurde auf eine ältere Karte abgestimmt, die keine Jahresinformation abgibt. ProDOS kalkuliert das Jahr aus dem Tagesdatum und dem Wochentag.

Um ProDOS-kompatibel zu sein, darf die Jahresinformation aus der N-CLOCK-Karte nicht übernommen werden. Die ProDOS-Zeittabelle reicht von 1981 bis 1987. Nach 1987 erhalten Sie erneut 1981.

Obwohl die N-CLOCK 1988 zur Verfügung stellt, liest ProDOS die Jahresinformation nicht. Spätere ProDOS-Versionen werden mit Sicherheit auch die Jahresangabe der N-CLOCK abfragen. Bitte nehmen Sie bis dahin dieselben Einschränkungen hin, mit denen die Besitzer anderer Karten für immer leben müssen.

## CP/M UND DIE N-CLOCK

Die N-CLOCK kann mit der geeigneten Software unter CP/M benutzt werden. Wenn Sie GBASIC oder MBASIC einsetzen, besteht die einfachste Möglichkeit darin, das unten aufgeführte Beispielprogramm einzusetzen.

```
010 DATA 169,0,133,156,133,54,169,196,133,57,133,55,169,58,133,51,32,106,253,96
020 REM
030 REM PUFFER AUF ADRESSE $200. ABGELEGT UNTER M MIT SINGLE PRECISION.
035 REM ZEIT = HEXADEZIMALE KONSTANTE AUF $240. KARTE IN SLOT 4
040 REM
050 M=61952 !:MP=M+64:ZEIT=&H240:SLOT=4
060 FOR I=MP TO MP+19
070 READ D:POKE I,D:
080 NEXT I
090 POKE MP+7,192+SLOT:POKE MP+13,ASC(" ")
100 CALL %TIME
110 FOR I=M TO M+31
120 A=PEEK(I)-128
130 IF A=34 THEN A=32
140 IF A <> 13 THEN A$ = A$+CHR$(A):
150 NEXT I
160 PRINT A$
170 END
```

## N-CLOCK-UHRKARTE

## LESEN DER N-CLOCK UNTER MASCHINENSPRACHE

Anwender, die mit Maschinensprache arbeiten, können die Routinen der eingebauten Betriebssoftware nutzen. Wenn Sie den N-CLOCK-Modus gewählt haben (Schalter 3 und 4 OFF), stehen Ihnen zwei Eingangsadressen in die Betriebssoftware zur Verfügung.

Nach Ausführung dieser kurzen Routine steht die Zeichenkette mit der Zeit und dem Datum im Pufferbereich ab \$200. Sie ist abgeschlossen mit einem Wagenrücklauf.

```
0300 A9 BA 0000 LDA #": " N-CLOCK BETRIEBSMODUS ANWÄHLEN
0302 20 0B C4 0010 JSR $C40B AUSWAHL DES MODUS
0305 20 08 C4 0020 JSR $C408 ZEIT LESEN-AUSGABE AUF $200
0308 60 0030 RTS
```

Wenn die N-CLOCK nicht in Platz 4 steckt, modifizieren Sie bitte Zeile 0010 und 0020 entsprechend. Der Eingang auf Adresse \$Cs0B empfängt das Format-Auswahl-Zeichen. Der Eingang auf Adresse \$Cs08 liest die Uhr und speichert die korrekt formatierte Zeichenkette unter Adresse \$200 ab. Die Zeichenkette wird durch einen Wagenrücklauf abgeschlossen.

Wenn Sie ein Programm in Maschinensprache verfassen wollen, welches die Uhr unabhängig von Dip-Switch-Schalterstellung liest, können Sie Adresse \$CsFF auswerten, um das Ausgabeformat zu ermitteln.

```
0300 A9 C4 0010 LDA #$C4 WIRD VOM AUFRUFENDEN PROGRAMM
BENUTZT, WENN EIN UNTERSCHIED VOR-
HANDEN IST. SLOT 4 = $C4.
0302 8D 10 03 0020 STA SLOT1
0305 8D 19 03 0040 STA SLOT2
0309 8D 1C 03 0050 STA SLOT3
030B 8D 25 03 0060 STA SLOT4
030E AD FF C4 0070 LDA $C4FF N-CLOCK MODUS. SCHALTER 3+4 OFF ?
0310 0080 SLOT1.EQ *-1 $ CN BYTE
0311 C9 03 0090 CMP #$03
0313 D0 09 0100 BNE KEIN N-CLOCK-MODUS
0315 A9 BA 0110 LDA #": " N-CLOCK-BETRIEBSART AUSWÄHLEN
0317 20 0B C4 0120 JSR $C40B SELEKTION DES MODUS
0319 0130 SLOT2.EQ *-1 $CN BYTE
031A 20 08 C4 0140 JSR $C408 ZEIT AUF ADRESSE $200 EINSETZEN
031C 0150 SLOT3.EQ *-1
031D 60 0160
0165 ALTER MODUS
031E A5 38 0170 LDA $38 KSWL AKTUELLE EINGANGSADRESSE
0320 48 0180 PHA AUF DEN STAPEL LEGEN
0321 A5 39 0190 LDA $39 KSWH
0323 48 0200 PHA
```



## N-CLOCK-UHRKARTE

```

0324 A9 C4      0210   LDA #C4  $CN N=SLOT, IN DEM DIE CLOCK STECKT
0325          0220 SLOT4 .EQ *-1
0326 85 39      0230   STA $39  KSWH
0328 A2 00      0240   LDX#0
032A 86 38      0250   STX $38  KSWL
032C 20 18 FD   0260 .1   JSR $FD18 MONITOR EINGABE ROUTINE
032F 9D 00 02  0270   STA $200,X ZEICHEN DER UHR ABSPEICHERN
0332 E8         0280   INX
0333 C9 8D      0290   CMP #$8D ENDE DER DATEN ?
0335 D0 F5      0300   BNE .1

0337 68         0310   PLA     EINGANGSADRESSE WIEDER EINSETZEN
0338 85 39      0320   STA $39  KSWH
033A 68         0330   PLA
033B 85 38      0340   STA $38  KSWL
0330 60         0350   RTS

```

## LESEN DER N-CLOCK UNTER UCSD-PASCAL

Für das UCSD-Pascal-Betriebssystem finden Sie nachstehend einige Programme, um die N-CLOCK im Rahmen eines PASCAL-Programmes zu benutzen.

Die folgende Funktion LESENCLOCK muß mit dem UCSD-Assembler assembliert und später an die N\_CLOCK Unit gelinkt werden.

```

;MACRO POPS 16 BIT ARGUMENT
;

```

```

        .MACRO    POP
        PLA
        STA      %1
        PLA
        STA      %1+1
        .ENDM

```

```

;MACRO PUSH 16 BIT ARGUMENT
;

```

```

        .MACRO    PUSH
        LDA      %1+1
        PHA
        LDA      %1
        PHA
        .ENDM

```

## N-CLOCK-UHRKARTE

```

;MACRO DUMPED 2 WORTE (NULL)
;

```

```

        .MACRO    DUMP
        PLA
        PLA
        PLA
        PLA
        .ENDM

```

```

;FUNCTION, UM DIE N-CLOCK ZU LESEN
;

```

```

        .FUNC      LESENCLOCK, 1

RETURN    .EQU      0      ;RETURN-ADRESSE UNTER $00 ABLEGEN
HOLD      .EQU      10     ;DATEN, UM DIE UHR ANZUHALTEN
RESTART   .EQU      2F     ;DATEN, UM DIE UHR WEITERLAUFEN ZU
                        ;LASSEN
CONTRG    .EQU      04     ;CONTROLREGISTER-DATENINIT
PORTA     .EQU      0C080  ;PORT A BASISADRESSE
CRTA      .EQU      0C081  ;CONTROLREGISTER A - BASIS ADRESSE
PORTB     .EQU      0C082  ;PORT B BASISADRESSE
CRTB      .EQU      0C083  ;CONTROLREGISTER B - BASIS ADRESSE
          .PUBLIC    SLOT  ;SLOT AUS DEM PASCAL-PROGRAMM
          POP        RETURN
          DUMP
          PLA        ;LSB VON SELECT
          AND        #0F   ;OBERES NIBBLE AUSBLENDEN
          TAY        ;Y FÜR DIE INDIZIERUNG EINSETZEN
          LDX        SLOT
          LDA        #00
          STA        CRTA,X ;DATUMRICHTUNGSREGISTER WÄHLEN.
          STA        CRTB,X ;
          STA        PORTA,X ;PORT A AUF EINGABE STELLEN
          LDA        #0FF
          STA        PORTB,X ;PORT B AUF AUSGABE STELLEN
          LDA        #CONTRG ;FÜR DATENAUSTAUSCH VORBEREITEN
          STA        CRTA,X
          STA        CRTB,X
          JSR        LESECL ;ZIFFER EINLESEN
          CPY        #00   ;PRÜFEN, OB ENDE
          BEQ        ENDE  ;WENN JA, DANN ENDE
          STA        TEMP  ;MIT 10 MULTIPLIZIEREN
          ASL        A
          ASL        A
          CLC
          ADC        TEMP
          ASL        A
          STA        TEMP
          INY        ;NÄCHSTE ZIFFER EINLESEN

```



## N-CLOCK-UHRKARTE

```

JSR      LESECL
CLC
ADC      TEMP      ;ADDIERE ZUR 10ER-STELLE
ENDE    PHA
LDA      #RESTART
STA      PORTB,X
PUSH    RETURN    ;RETURN ADRESSE WIEDERHERSTELLEN
RTS     ;RÜCKKEHR ZU PASCAL
LESECL LDA      #HOLD ;START-UHR LESEN
STA      PORTB,X
LDA      TABLE,Y ;CODE FÜR EINE ZIFFER HOLEN
STA      PORTB,X ;CODE SENDEN
NOP     ;WARTEN
NOP     ;WARTEN
LDA      PORTA,X ;DATEN HOLEN
CPY     #03      ;TAG HI ?
BEQ     MASK     ;WENN JA - LÖSCHEN
CPY     #07      ;STUNDEN HI ?
BNE     NOMASK   ;WENN NEIN, NICHT MASKIEREN
MASK    AND     #03 ;NUR BIT 0 UND 1 SICHERN

NOMASK  AND     #0F ;HIGH-BITS ABSCHNEIDEN
RTS

TABLE  .BYTE   36 ;CODE FÜR DEN WOCHENTAG
        .WORD  393A ;MONAT
        .WORD  3738 ;TAG
        .WORD  3B3C ;JAHR
        .WORD  3435 ;STUNDEN
        .WORD  3233 ;MINUTEN
        .WORD  3031 ;SEKUNDEN
TEMP   .BYTE   00 ;ZWISCHENSPEICHER FÜR MULT

        .END     ;ENDE

```

Die folgende Unit N\_CLOCK benutzt die externe Function LESENCLOCK, die oben aufgeführt ist. Nach der Compilierung dieser Unit muß LESENCLOCK gelinkt werden.

UNIT N\_CLOCK; INTRINSIC CODE 26 DATA 27;

(\*-----\*)

## INTERFACE

```

FUNCTION UWOCHENTAG :INTEGER;
FUNCTION UMONAT :INTEGER;
FUNCTION UMONATSDATUM :INTEGER;
FUNCTION UJAHR :INTEGER;
FUNCTION USTUNDE :INTEGER;
FUNCTION UMINUTE :INTEGER;
FUNCTION USEKUNDE :INTEGER;

```

## N-CLOCK-UHRKARTE

## IMPLEMENTATION

## TYPE

```

BYTE = PACKED ARRAY[0..1] OF 0..255;
TTRICK = RECORD CASE BOOLEAN OF
    TRUE: (INT: INTEGER);
    FALSE: (PTR: ^BYTE)
END;

```

## VAR

SLOT, X: INTEGER;

(\* DIE FOLGENDE FUNKTION IST DIE OBEN AUFGEFÜHRTE ASSEMBLERROUTINE \*)

FUNCTION LESENCLOCK (SELECT: INTEGER): INTEGER; EXTERNAL;

FUNCTION UWOCHENTAG;

```

BEGIN
    UWOCHENTAG := LESENCLOCK (0)
END;

```

FUNCTION UMONAT;

```

BEGIN
    UMONAT := LESENCLOCK (1)
END;

```

FUNCTION UMONATSDATUM;

```

BEGIN
    UMONATSDATUM := LESENCLOCK (3)
END;

```

FUNCTION UJAHR;

```

BEGIN
    UJAHR := LESENCLOCK (5)
END;

```

FUNCTION USTUNDE;

```

BEGIN
    USTUNDEN := LESENCLOCK (7)
END;

```

FUNCTION UMINUTE;

```

BEGIN
    UMINUTEN := LESENCLOCK (9)
END;

```

FUNCTION USEKUNDE;

```

BEGIN
    USEKUNDE := LESENCLOCK (11)
END;

```



## N-CLOCK-UHRKARTE

```

FUNCTION PEEK (ADDR: INTEGER): INTEGER;
VAR TRICK: TTRICK;
BEGIN
  WITH TRICK DO
    BEGIN
      INT := ADDR; PEEK := PTR^[0]
    END
  END;

```

(\*-----\*)

```

BEGIN (* HAUPT-UND INITIALISIERUNGSTEIL DER UNIT *)
(* DIESES PROGRAMMTEIL DURCHSUCHT ALLE SLOTS DARAUf, OB EINE N-CLOCK
EXISTENT IST. WIRD DIE KARTE GEFUNDEN, ÜBERGIBT DIE PROZEDURE DIESE INFOR-
MATION AN DIE ASSEMBLER-ROUTINE. WIRD KEINE UHR GEFUNDEN, WIRD DIE PRO-
GRAMMAUSFÜHRUNG ABGEBROCHEN. *)

```

```

X := PEEK (-12289);      (* CFFF STROBE *)
SLOT := 0;
FOR X := 1 TO 7 DO
  BEGIN
    IF (
      (PEEK (-16350 + X * 256) = 248)
      AND (PEEK (-16348 + X * 256) = 104)
    )
    OR (
      (PEEK (-16384 + X * 256) = 8)
      AND (PEEK (-16383 + X * 256) = 120)
      AND (PEEK (-16130 + X * 256) = 178)
    ) THEN SLOT := X * 16;
    X := PEEK (-12289)      (* CFFF STROBE *)
  END;
IF SLOT = 0 THEN
  BEGIN
    PAGE (OUTPUT); WRITELN ('Die N-CLOCK ist nicht eingesetzt !!'); EXIT (PROGRAM)
  END
END.

```

## N-CLOCK-UHRKARTE

Das folgende Beispielprogramm setzt automatisch das Datum im UCSD-PASCAL-System ein.

```

PROGRAM DATUM;
USES N_CLOCK;

```

```

TYPE
DATUM = PACKED RECORD
  MONAT : 1..12;
  TAG   : 1..31;
  JAHR  : 0..99;
END;

```

```

VAR
BLOCK : ARRAY [0..255] OF DATUM;

```

```

PROCEDURE POKE (ADRESSE:INTEGER; WERT:DATUM);

```

```

VAR
MEMREF : RECORD CASE INTEGER OF
  1: (ADDR:INTEGER);
  2: (PTR:^DATUM);
END;

```

```

BEGIN
MEMREF.ADDR:=ADRESSE; MEMREF.PTR^:=WERT;
END;

```

```

BEGIN
UNITREAD (4,BLOCK,512,2);
  WITH BLOCK[10] DO
    BEGIN
      TAG:= UMONATSDATUM; MONAT:= UMONAT; JAHR:=UJAHR;
    END;
UNITWRITE (4,BLOCK,512,2); POKE (-21992, BLOCK[10]);
PAGE (OUTPUT);WRITELN('Heute ist der ',DATUM);
END.

```



## N-CLOCK-UHRKARTE

## LESEN DER N-CLOCK OHNE NUTZUNG DER BETRIEBS SOFTWARE

Wenn Sie die N-CLOCK unter Umgehung der Betriebssoftware direkt lesen wollen, müssen Sie prinzipiell folgende Schritte programmieren:

1. Den PIA-Chip initialisieren.
2. Die N-CLOCK anhalten.
3. Eine Ziffer adressieren.
4. Die Ziffer lesen.
5. Position 3 und 4 wiederholen, bis alle relevanten Ziffern gelesen sind.
6. Die N-CLOCK freigeben und Interrupts zulassen.

Das folgende Programm zeigt beispielhaft die Programmierung. Es ist tatsächlich nur als Muster brauchbar, da Applesoft-BASIC nicht schnell genug ist, um den Update der Zeit im Sekundenintervall vorzunehmen. In dem Programm wird die PIA durch die Subroutine auf 130 initialisiert. Das Unterprogramm ab Zeile 322 liest die Formatdefinition in den Bereich FMT ein. Die Zahlen in der Liste sind entweder die Adressen der Ziffern (Werte zwischen 48 und 60) oder ASCII Zeichen > 127.

## REGISTER DER UHR

	ZEHNER	EINER
JAHR	60	59
MONAT	58	57
TAG IM MONAT	56	55
WOCHENTAG	-	54
STUNDE	53	52
MINUTE	51	50
SEKUNDE	49	48

## ASCII ZEICHEN DEZIMALE WERTE

LEERZEICHEN	160
SCHRÄGSTRICH	175
DOPPELPUNKT	186
FORMATENDE	0

Das Unterprogramm ab Zeile 214 geht schrittweise durch den Format-Bereich und liest die Uhr. In Zeile 250 wird die Uhr angehalten, in Zeile 300 wieder freigegeben.

Das Basic-Programm ist von der Konzeption her dem Assembler-Programm der N-CLOCK sehr ähnlich. Durch Variationen der Formate können die unterschiedlichsten Arrangements der Uhr-Daten angezeigt werden.

## N-CLOCK-UHRKARTE

```

010 REM DATENBEREICH DIMENSIONIEREN UND DEN STECKPLATZ DER N-CLOCK
014 REM FESTLEGEN.
020 DIM FMT(30):SLOT=4
032 REM DATENFORMAT LESEN
040 GOSUB 322
052 REM PIA INITIALISIEREN
060 GOSUB 130
072 REM UHR LESEN
080 GOSUB 214
090 VTAB 10, HTAB 10:PRINT T$:IF PEEK(-16384) <128 THEN 72
100 POKE -16368,0
110 END
120 REM *****
130 REM PIA INITIALISIEREN.
135 REM $C080 + $N0
150 PA = 49280 + SLOT * 16: CA = PA + 1: PB = PA + 2: CB = PA + 3
156 REM WENN PIA SCHON INITIALISIERT IST, DANN 200
160 IF PEEK (CB) <> 0 THEN 200
170 REM
172 REM RICHTUNGSREGISTER DER PIA EINSTELLEN
175 REM
180 POKE CA,0:POKE CB,0: POKE PA,0: POKE PB, 255:
184 REM
186 REM AUF DIE DATENREGISTER ZEIGEN
188 REM
190 POKE CA,4: POKE CB,4:POKE PB,47
200 RETURN
210 REM *****
214 REM UHR LESEN UNTER BENUTZUNG DES FORMATS
220 T$=""
230 FOR I= 1 TO NC
240 REM
244 REM UHR ANHALTEN
250 POKE PB, 16: D= FMT(I): IF D > 127 THEN GOTO 285
260 M = 16: IF D = 56 OR D = 53 THEN M = 4
270 REM
274 REM ZIFFER LESEN
280 D = T - INT ( T / M ) * M + 176
285 T$ = T$ + CHR$ (D)
290 NEXT
300 POKE PB, 47
310 RETURN
320 REM
322 REM DATENFORMAT "T TA/MO/JA ST:MI:SE" LESEN
330 DATA 54, 160, 56, 55, 175, 58, 57, 175, 60, 59, 160, 53, 52, 186, 51, 50, 186, 49, 48, 0
340 NC = 0:
350 READ D: IF D = 0 THEN RETURN
360 NC = NC + 1: FMT (NC) = D: GOTO 350

```



## INTERRUPT-BETRIEB DER N-CLOCK

Eine der Eigenschaften der N-CLOCK-Uhrkarte ist die Möglichkeit, Interrupts in einem genau festgelegten zeitlichen Abstand zu generieren. Damit wird Vordergrund- und Hintergrundverarbeitung möglich. Auch können Daten in genau definierten Intervallen gelesen werden. Schalter 6 der N-CLOCK muß auf ON stehen.

Folgende Intervalle stehen zur Verfügung:

1/1024 Sek., eine Sekunde, eine Minute, eine Stunde. Diese Intervalle werden durch Abspeichern eines Wertes im Control-Register der N-CLOCK festgelegt.

Zeit	Control	Register
	CRA	CRB
-	\$ 04	\$ 04
1/1024 Sek	\$ 05	\$ 04
1 Sekunde	\$ 0C	\$ 04
1 Minute	\$ 04	\$ 05
1 Stunde	\$ 04	\$ 0C

Beachten Sie bitte, daß das 1/1024-Intervall 1/1024stel einer Sekunde ist, also keine Millisekunde, sondern ca. 0.977 Millisekunde. Programme, die echte Millisekunden benötigen, müssen den Wert des 1024 Hz-Zählers mit 1000/1024 multiplizieren.

Das folgende Basic-Programm zeigt die Benutzung von Interrupts ohne Assemblerprogrammierung. Es ist eingestellt auf einen Interrupt / Sek. Drei kleine Maschinenprogramme werden durch Poke gespeichert. Durch CALL 782 wird der Interrupt ermöglicht. Damit Applesoft einen Interrupt registriert, wird bei jedem Interrupt durch das Programm auf Adresse \$300 ein Wert <> 0 auf Adresse 779 abgelegt.

Das Programm bewegt sich in einer Schleife auf Zeile 600, bis ein Wert <> 0 (Interrupt) festgestellt wird. Bei jedem Schleifendurchlauf wird ein Punkt auf dem Bildschirm gedruckt. Sobald PEEK (77) endgültig <> 0 ist, wird Speicherstelle 77 auf 0 gesetzt, Variable C um 1 erhöht und ein Schrägstrich ausgegeben. Sobald C endgültig den Grenzwert T erreicht, liest das Programm die N-CLOCK und zeigt sie auf dem Bildschirm.

## BEISPIELPROGRAMM FÜR DEN INTERRUPT-BETRIEB DER UHRKARTE UNTER APPLESOFT-BASIC

```

007 REM KARTE STECKT IN SLOT 4
010 SLOT = 4
017 REM DIE FOLGENDEN 3 KLEINEN ASSEMBLER-ROUTINEN WERDEN DURCH
018 REM DATA/POKE GELADEN
020 REM 768 LDA #01 ;FLAG FÜR BASIC SETZEN
022 REM STA 779
024 REM LDA PIA ;PIA LÖSCHEN
026 REM LDA $ 45 ;AKKUMULATOR WIEDERHERSTELLEN
028 REM RTI ;RETURN
029 REM INTERRUPTS UNTERBINDEN
030 REM 779 BYTE 00 ;BASIC FLAG
032 REM 780 SEI
034 REM RTS
036 REM INTERRUPTS ERMÖGLICHEN
038 REM 782 CLI
040 REM RTS
050 DATA 169, 1, 141, 11,3 ,173 ,208, 192, 165, 69, 64, 0, 120, 96, 88, 96
060 FOR A= 768 TO 783: READ X: POKE A, X: NEXT A
080 REM INTERRUPTS AUSSCHALTEN
100 CALL 780
120 REM AUF DIE INTERRUPT-ROUTINE ZEIGEN
140 POKE 1022,0: POKE 1023,3
160 REM BASIS ADRESSE DER PIA $ C080+$N0
180 A = -16256 + 16 * SLOT
200 REM DATENRICHTUNGSREGISTER
220 POKE A + 1,0: POKE A + 3,0
240 REM PORT B AUSGABE, 8BIT
260 POKE A + 2, 255
280 REM DIE NÄCHSTEN BEIDEN POKES LEGEN DIE INTERRUPT-RATE FEST. BASIC IST
285 REM ZU LANGSAM FÜR 1024 HZ-INTERRUPTS.
310 REM CONTROL-REGISTER A AUF 1 INTERRUPT JE SEK. STELLEN.
330 POKE A + 1, 12
350 REM CONTROL-REGISTER B.
370 POKE A + 3, 4
390 REM DIE PIA SOLL DIE INTERRUPTS WEITERLEITEN
410 POKE A + 2, 47
430 REM FLAGS DER PIA LÖSCHEN
440 X = PEEK (A)
460 REM BASIC FLAG ADRESSE
470 F = 779
490 REM FLAG AUF 0. INTERRUPTS ERMÖGLICHEN. ANZAHL INTERRUPTS/ZYCLUS
500 Z = 0: CALL 782: T = 3
510 ONERR GOTO 570

```



## N-CLOCK-UHRKARTE

```

530 REM INITIALISIERUNG
540 GOTO 710
560 REM VOR DEM STOP INTERRUPTS ABSCHALTEN
570 CALL 780: STOP
590 REM AUF DEN INTERRUPT WARTEN. ANSONSTEN EINEN PUNKT DRUCKEN
600 IF PEEK (F) = Z THEN PRINT ".,."; GOTO 600
620 REM BEI JEDEM INTERRUPT EINEN / DRUCKEN.
630 PRINT "/";
650 REM NACH DEM INTERRUPT FLAG ERNEUT SETZEN
660 POKE F,Z
670 REM ZÄHLER HOCHSETZEN UND MIT MAXIMUM VERGLEICHEN
680 C = C + 1: IF C < T THEN GOTO 600
690 REM DIE FOLG. ANWEISUNGEN AUSFÜHREN, WENN DER ZÄHLER MAX.ERREICHT.
700 REM ZÄHLER INITIALISIEREN
710 C = 0:PRINT: PRINT: PRINT CHR $ (4) "IN # SLOT": PRINT CHR$ (4) "PR#" SLOT
720 VTAB 5: INPUT".,."; K$: PRINT CHR $ (4),"IN # 0"
730 PRINT CHR$(4); "PR#0": PRINT K$: CALL - 958
735 REM CONTROL-REGISTER AUF 1 INTERRUPT P/SEK
740 POKE A + 1, 12
750 REM CONTROL-REGISTER B
760 POKE A + 3,4: GOTO 600

```

## N-CLOCK-UHRKARTE

## DOS 3.3 IN VERBINDUNG MIT DEM INTERRUPT-BETRIEB

DOS 3.3 ist nicht für die Verarbeitung von Interrupts eingerichtet. Die Interrupt-Request-Routine innerhalb des Apple Monitor ROM's rettet das A-Register auf Adresse \$45 im RAM-Bereich. Dies wäre kein Problem, wenn DOS 3.3 Adresse \$45 nicht bei über 20 verschiedenen Gelegenheiten als temporäre Variable benutzen würde. Ein Interrupt zur falschen Zeit kann das DOS, die Diskette oder ein Programm stören.

Dieses Problem wäre einfach dadurch zu umgehen, wäre, den IRQ-Interrupt vor oder während einer DOS-Operation zu blockieren und danach wieder zu erlauben. Die beste Lösung allerdings ist es, DOS 3.3 so zu modifizieren, daß Adresse \$45 nicht mehr benutzt wird.

Das Basicprogramm DOSIRQ ermöglicht vollständige Verträglichkeit von DOS mit dem IRQ-Interrupt. Sollten Sie allerdings 1024/Sek.-Interrupts benötigen, können Sie immer noch in Konflikt mit DOS 3.3 geraten. Da es bei Lese- oder Schreiboperationen auf genaues Timing ankommt, unterbindet DOS 3.3 den IRQ-Interrupt. Während der Motor des Laufwerkes hochgefahren wird, Spur und Sektor gesucht und Daten gelesen oder geschrieben werden, wird der IRQ-Interrupt unterbunden.

```

10 HOME:PRINT"DOS 3.3 MODIFIKATION INTERRUPT"
20 POKE 49061, 70: POKE 49059, 70: POKE 49053, 70: POKE 48987,70: POKE 48983,70
30 POKE 48981, 70: POKE 48953, 70: POKE 48918, 70: POKE 48851,70: POKE 44632,70
40 POKE 44628, 70: POKE 44554, 70: POKE 44474, 70: POKE 42879,70: POKE 42855,70
50 POKE 41676, 70: POKE 41473, 70: POKE 41465, 70: POKE 41463,70: POKE 41448,70
60 POKE 41427, 70: POKE 41406, 70: POKE 41304, 70: POKE 41278,70: POKE 41267,70
70 POKE 48548, 44: POKE 47622, 44
80 END

```

## INTERRUPTS VERHINDERN

Der IRQ-Interrupt kann durch die SEI-Instruktion unterbunden werden. Der IRQ-Interrupt wird unterbunden, wenn die RESET-oder CTRL- RESET-Taste des Apple-Computers betätigt wird. RESET setzt zusätzlich die Interrupt-Auswahlen der N-CLOCK Karte zurück, sodaß sowohl IRQ- als auch NMI-Interrupts beendet werden.

Sowohl der IRQ- als auch der NMI-Interrupt der N-CLOCK können durch Speicherung des Wertes 0 in den Control-Registern (\$ C081 + n0 und \$ C083 + n0 ) gelöscht und unterbunden werden. Wenn die Schalter 5 + 6 der DIP-Switch-Gruppe der N-CLOCK auf OFF stehen, liegen die Interrupts nicht mehr auf dem Systembus.



N-CLOCK-UHRKARTE

ZEIT UND DATUM WÄHREND DES INTERRUPT-BETRIEBES  
ERMITTELN

Es ist nicht zweckmäßig, das Betriebsprogramm der N-CLOCK zu benutzen, um die Uhrzeit und das Datum zu lesen, wenn ein Programm eingesetzt wird, das Interrupts benutzt. Das Betriebsprogramm der N-CLOCK benutzt den System-Eingabepuffer auf Adresse \$ 200, um den Zeit/Datum-String aufzubauen. Im N-CLOCK-Betriebsmodus wird ein Teil des Betriebsprogrammes vom ROM der Karte in das RAM des Apple II auf Adresse \$ 220 kopiert. Die beste Lösung ist es, ein separates Maschinenprogramm zum Lesen der Uhrkarte zu benutzen, da die Benutzung des System-Eingabepuffers zu Konflikten mit Programmen führt, die im Vordergrund ablaufen.

Sollte der Clock-Interrupt nicht alle Daten der Uhr benötigen, könnte ein separates Maschinenprogramm, das nur einige Datenregister der Uhr liest, u.U. schneller arbeiten als bei dem Versuch, die ganze Uhrzeit unter Verwendung der Betriebssoftware zu lesen.

Das Betriebsprogramm büßt oftmals die Angaben über die Interrupt-Intervalle ein. Wenn Sie nicht nach jedem Lesen der Uhr die Intervalldauer neu programmieren wollen, sollten Sie ein separates Programm zum direkten Lesen der Uhr installieren.